MTH1030 PROJECT 1

ZHIQI FANG 35954396

Problem 1: The cube

1. Problem Restatement

We are given the coordinates of two opposite corners of the top face of a cube :

$$A = (84.5744, 46.5037, 11.5858) \quad \text{and} \quad C = (68.2693, 11.5320, 22.3241)$$

The cube is partially sunk into the ground (the $x\,y$ - plane, $z = 0$). The visible top corners are A, B, C, D, and their corresponding ground points are A', B' C', D' connected by the cube edges AA', BB', CC', DD'. It is given that B and D are at the same height above the ground .

The task is to determine:

(a) The coordinates of A, B, C, D and A', B' C', D'

(b) The lengths $|AA'|$, $|BB'|$, $|CC'|$, $|DD'|$

(c) The side length $|AB|$

(d) The area of A', B' C', D'

(e) The surface area of the part of the cube above the ground

(f) The volume of the part of the cube above the ground

(g) Produce a 3D model of the structure (using vector methods)

2. Methodology and Calculations

(1) Determining Points B and D

Since A and C are opposite vertices of the square, their midpoint centre O, is also the midpoint of diagonal BD.

Let $O=\frac{A+C}{2}$, vector $u = A - O$. We seek a vector v such that $B=O+v$ and $D=O-v$.

Then vector v must satisfy: 1. $v \perp u$ 2. $|v| = |u|$ 3. $v_z=0$ (Since B and D are at the same height)

In the code:

pA, **pB**, **pC**, and **pD** denote the position vectors of the top vertices, **pO** the centre of the square, **u** the vector from the centre to A, **ux** and **uy** its horizontal components, **normU** its magnitude, **normUxy** the magnitude of its projection onto the x y-plane, and **v** the vector from the centre to B.

In[1]:=
```
ClearAll["Global`*"];
pA = {84.5744, 46.5037, 11.5858};
pC = {68.2693, 11.5320, 22.3241};
pO = (pA + pC)/2;
u = pA - pO;
ux = u[[1]];uy = u[[2]];
normU = Norm[u];
normUxy = Sqrt[ux^2 + uy^2];
v = (normU/normUxy) {-uy, ux, 0};
pB = pO + v
pD = pO - v
```

Out[10]=
```
{58.2715, 37.4802, 16.955}
```

Out[11]=
```
{94.5722, 20.5555, 16.955}
```

We now check that the four top points really form a square . For a square, adjacent sides must have equal length and be perpendicular .

In[12]:=
```
AB=pB-pA;
BC=pC-pB;
CD=pD-pC;
DA=pA-pD;
checkLengths={Norm[AB],Norm[BC],Norm[CD],Norm[DA]};
checkRightAngles={AB.BC,BC.CD,CD.DA,DA.AB};
{checkLengths,checkRightAngles}//N
```

Out[18]=

$$\{\{28.3213, 28.3213, 28.3213, 28.3213\}, \{2.41585\times10^{-13}, -2.52243\times10^{-13}, -1.56319\times10^{-13}, 2.23821\times10^{-13}\}\}$$

Since all four side lengths are equal and all adjacent dot products are zero, the points A , B , C , D do indeed form a square.

(2) Determining the direction of the vertical edges

The edges AA’, BB’, CC’ and DD’ are perpendicular to the top face ABCD. Therefore the direction of these edges can be obtained from a normal vector to the plane ABCD, found using a cross product.

In the code, **AB** and **AD** denote the vectors from vertex A to vertices B and D respectively, and **edgeDir** denotes the unit normal vector to the top face ABCD, obtained using the cross product, which determines the direction of the cube edges perpendicular to this face .

In[19]:=
```
AB = pB - pA;
AD = pD - pA;
edgeDir = Normalize[Cross[AD,AB]];
edgeDir//N
```

Out[22]=
```
{-0.113293, -0.242994, -0.963389}
```

(3) Edge lengths |AA'|, |BB'|, |CC'|, |DD'|

Each ground point lies on the line through the corresponding top vertex in the cube-edge direction. For a general top vertex P∈{A,B,C,D}, the edge line is:

$$r_p(t) = P + t\,e$$

where e = *edgeDir*

Since the ground is the plane $z = 0$, the parameter t for each vertex is determined by setting the third coordinate of $r_p(t)$ equal to zero.

This gives $P_Z + t\,e_Z = 0 \Rightarrow t = -\frac{P_Z}{e_Z}$

The same formula is applied to vertices A,B,C, and D. Because edgeDir is a unit vector, the length of each exposed vertical edge segment is simply the absolute value of the corresponding parameter.

Hence,|AA'|=| t_A |, |BB'|=| t_B |, |CC'|=| t_C |, |DD'|=| t_D |

In the code, **tA**, **tB**, **tC**, and **tD** denote the intersection parameters for vertices A, B, C, and D.

In[23]:=
```
tA=-pA[[3]]/edgeDir[[3]]
tB=-pB[[3]]/edgeDir[[3]]
tC=-pC[[3]]/edgeDir[[3]]
tD=-pD[[3]]/edgeDir[[3]]
```

Out[23]=

12.0261

Out[24]=

17.5993

Out[25]=

23.1725

Out[26]=

17.5993

(4) Intersection points A', B', C', D'

Each ground point is obtained by substituting the corresponding parameter into the parametric edge equation:

$$r_p(t) = P + t\,e$$

where e = edgeDir

Thus, once the parameters t_A , t_B , t_C , t_D have been determined, the ground intersection points are given by

$$A' = A + t_A\,e, \quad B' = B + t_B\,e, \quad C' = C + t_C\,e, \quad D' = D + t_D\,e$$

This corresponds to translating each top vertex along the cube-edge direction until it intersects the ground plane z=0.

In the code, **pAprime**, **pBprime**, **pCprime**, and **pDprime** denote the position vectors of the ground intersection points.

In[27]:=
```
pAprime=pA+tA edgeDir;
pBprime=pB+tB edgeDir;
pCprime=pC+tC edgeDir;
pDprime=pD+tD edgeDir;
{{"A'", N[pAprime, 6]},{"B'", N[pBprime, 6]}, {"C'", N[pCprime, 6]},{"D'", N[pDprime, 6]} }/
```

Out[31]=

$\{\{A', \{83.2119, 43.5814, 0.\}\}, \{B', \{56.2776, 33.2037, 0.\}\},$
$\{C', \{65.644, 5.90123, 3.55271\times10^{-15}\}\}, \{D', \{92.5783, 16.279, 0.\}\}\}$

The small non - zero value 3.55271×10^{-15} is due to floating - point round - off error in numerical computation and is therefore treated as zero .

(5) Side length of the cube

Since *A* and *B* are adjacent vertices of the square top face ABCD, the cube side length is given by |AB|=‖B−A‖.

In the code, **side** denotes the side length AB

In[32]:=
```
side = Norm[pB - pA]
```

Out[32]=

28.3213

Thus, side length = 28.32 m.

(6) Area of the Base A' B' C' D'

The base A' B' C' D' lies in the x y - plane.Since it is obtained by projecting the square ABCD along the cube-edge direction onto the ground plane. Hence it forms a parallelogram. The area of this parallelogram is given by the magnitude of the cross product of two adjacent side vectors:

$$\textbf{Area}\,(A'\,B'\,C'\,D') = \|\,(B' - A')\times(D' - A')\,\|$$

In the code, this is computed as :

In[33]:=
```
baseArea=Norm[Cross[pBprime-pAprime,pDprime-pAprime]];
N[baseArea]
```

Out[34]=

832.575

Thus, base area = 832.58 m^2

(7) Surface Area Above Ground

We include only the faces exposed above the plane z=0. Therefore, we exclude the ground face A', B' C', D', since it lies entirely in the ground plane.
Each visible side region above the ground is a planar quadrilateral.We compute each quadrilateral area by splitting it into two triangles and applying the cross product formula to each triangle.
The surface area above ground is therefore the sum of the top-face area and the four exposed side-region areas, excluding the ground face A', B' C', D' .

In[35]:=
```
areaTop=Norm[Cross[pB-pA,pD-pA]];

areaSideAB=1/2 Norm[Cross[pB-pA,pBprime-pA]]+1/2 Norm[Cross[pBprime-pA,pAprime-pA]];
areaSideBC=1/2 Norm[Cross[pC-pB,pCprime-pB]]+1/2 Norm[Cross[pCprime-pB,pBprime-pB]];
areaSideCD=1/2 Norm[Cross[pD-pC,pDprime-pC]]+1/2 Norm[Cross[pDprime-pC,pCprime-pC]];
areaSideDA=1/2 Norm[Cross[pA-pD,pAprime-pD]]+1/2 Norm[Cross[pAprime-pD,pDprime-pD]];

surfaceAreaAboveGround=areaTop+areaSideAB+areaSideBC+areaSideCD+areaSideDA
```

Out[40]=
```
2795.83
```

Thus, surface area=2795.83 m^2

(8) Volume Above Ground

To compute the volume above the ground, we duplicate the solid above the plane z=0 and place a congruent copy in the opposite orientation.Because the top surface is planar and the four edge vectors AA', BB', CC', DD' are parallel, the two congruent solids fit together to form a parallelepiped.
The base of this parallelepiped is the parallelogram $A'B'C'D'$ whose area vector is $C'B' \times C'D'$
Its side vector is $AA'+CC'$ which is equal to $BB'+DD'$
Therefore, the volume of the parallelepiped is

$$V_{para}=|(C'B' \times C'D') \cdot (AA'+CC')|$$

Since the original solid is exactly one half of this parallelepiped, its volume is

$$V=\frac{1}{2}|(C'B' \times C'D') \cdot (AA'+CC')|$$

In[41]:=
```
volumeAboveGround = Abs[
   Cross[pBprime - pCprime, pDprime - pCprime].((pA - pAprime) + (pC - pCprime))
]/2 ;
NumberForm[volumeAboveGround, {Infinity, 2}]
```

Out[42]//NumberForm=
```
14116.26
```

Therefore, the volume of the part of the cube above the ground is 14116.26 m^3

(9) 3D Representation

```
Graphics3D[{{Opacity[0.55], LightBlue, Polygon[{pA, pB, pC, pD}],
    Polygon[{pA, pB, pBprime, pAprime}], Polygon[{pB, pC, pCprime, pBprime}],
    Polygon[{pC, pD, pDprime, pCprime}], Polygon[{pD, pA, pAprime, pDprime}]},
   {Opacity[0.25], Gray, Polygon[{pAprime, pBprime, pCprime, pDprime}]},
   {Black, Thick, Line[{pA, pB, pC, pD, pA}],
    Line[{pAprime, pBprime, pCprime, pDprime, pAprime}], Line[{pA, pAprime}],
    Line[{pB, pBprime}], Line[{pC, pCprime}], Line[{pD, pDprime}]},
   {Red, PointSize[0.02], Point /@ {pA, pB, pC, pD, pAprime, pBprime, pCprime, pDprime}},
   {Black, Text[Style["A", 14, Bold], pA + {1, 1, 1}],
    Text[Style["B", 14, Bold], pB + {1, 1, 1}],
    Text[Style["C", 14, Bold], pC + {1, 1, 1}], Text[Style["D", 14, Bold], pD + {1, 1, 1}],
    Text[Style["A'", 14, Bold], pAprime + {1, 1, 1}], Text[Style["B'", 14, Bold],
     pBprime + {1, 1, 1}], Text[Style["C'", 14, Bold], pCprime + {1, 1, 1}],
    Text[Style["D'", 14, Bold], pDprime + {1, 1, 1}]}}, Axes → True,
 AxesLabel → {"x", "y", "z"}, Boxed → False, ImageSize → Large]
```

Out[ ]=



3. Summary

In[43]:=

```
fmt[x_] := NumberForm[N[x, 10], {Infinity, 2}];
summaryTable = Grid[
        {{"Point", "Coordinates"},
          {"A", pA},
          {"B", pB},
          {"C", pC},
          {"D", pD},
          {"A'", pAprime},
          {"B'", pBprime},
          {"C'", pCprime},
          {"D'", pDprime},
          {"|AA'|", Row[{fmt[Norm[pA - pAprime]], " m"}]},
          {"|BB'|", Row[{fmt[Norm[pB - pBprime]], " m"}]},
          {"|CC'|", Row[{fmt[Norm[pC - pCprime]], " m"}]},
          {"|DD'|", Row[{fmt[Norm[pD - pDprime]], " m"}]},
          {"Side length |AB|", Row[{fmt[side], " m"}]},
          {"Area A'B'C'D'", Row[{fmt[baseArea], " m^2"}]},
          {"Surface area above ground", Row[{fmt[surfaceAreaAboveGround], " m^2"}]},
          {"Volume above ground", Row[{fmt[volumeAboveGround], " m^3"}]}},
        Frame → All, Alignment → Left, Spacings → {2, 1},
        ItemStyle → {{Directive[Bold, 14]}, None}, Background → {None, {LightGray, None}}];
summaryTable
```

Out[45]=

Point	Coordinates
A	{84.5744, 46.5037, 11.5858}
B	{58.2715, 37.4802, 16.955}
C	{68.2693, 11.532, 22.3241}
D	{94.5722, 20.5555, 16.955}
A'	{83.2119, 43.5814, 0.}
B'	{56.2776, 33.2037, 0.}
C'	$\{65.644, 5.90123, 3.55271\times10^{-15}\}$
D'	{92.5783, 16.279, 0.}
|AA'|	12.03 m
|BB'|	17.60 m
|CC'|	23.17 m
|DD'|	17.60 m
Side length |AB|	28.32 m
Area A'B'C'D'	832.57 m^2
Surface area above ground	2795.83 m^2
Volume above ground	14116.26 m^3

Problem 2

1. Problem Statement

In this project we analyse linear systems arising from population transport between parallel planets . The first part studies transport from Earth 1 to Earth 2, and the second part considers an analogous model for Mars .

Earth part :

We are given an initial weight distribution $w \in R^5$ and an observed distribution $\tilde{w}$ after transport . The transport device is described by five column vectors u_1, u_2, u_3 , u_4, u_5 and tuning parameters

$$\alpha = \begin{pmatrix} a_1 \\ a_2 \\ a_3 \\ a_4 \\ a_5 \end{pmatrix}.$$

The task is to determine :

(a) Identify the matrix A as U diag (w)

(b) Determine whether the system is consistent and whether the solution is unique

(c) Compute the realised tuning parameters β using $\tilde{w}$

(d) Compute the intended tuning parameters θ that give $\tilde{w}$ = w

(e) Replace u_1 with a new vector v_1 so that with β the output becomes w

MAR Part

(f) consider an analogous transport problem on Mars with three components . Analyse three possible actions to achieve $\tilde{\omega} = \omega$

(i) adjust the tuning parameters

(ii) redistribute the population

(iii) rebuild the transport device

(g) Study a new Martian device and determine which candidate vector $\boldsymbol{y^4}$ yields a consistent system

2. Methodology and Calculations

Earth part

(1) Constructing the matrix A

The system can be written as the following two forms :

$$\tilde{w} = A\,\alpha$$

$$\tilde{w}_i = a_1 w_1 u_{i1} + a_2 w_2 u_{i2} + a_3 w_3 u_{i3} + a_4 w_4 u_{i4} + a_5 w_5 u_{i5}$$

so the j-th column of A is $w_j\, u_j$. If the columns of U=[u_1, u_2, u_3 , u_4, u_5] ,then A=[$w_1\, u_1\, w_2\, u_2\, w_3\, u_3\, w_4\, u_4\, w_5\, u_5$] , so A=U $DiagonalMatrix[w]$

In the code, **DW** denotes the diagonal matrix formed from the vector w, DW=diag(w).

In[181]:=

```
w = {0.60, 0.17, 0.13, 0.09, 0.01};
wt = {0.39, 0.22, 0.14, 0.05, 0.20};

u1 = {1, 5, 3, -5, 2};
u2 = {1, -3, 5, 2, 2};
u3 = {4, 1, 2, 2, -3};
u4 = {1, -3, 5, -2, 4};
u5 = {2, 4, -1, 2, -4};

U = Transpose[{u1, u2, u3, u4, u5}];
DW = DiagonalMatrix[w];

A = U . DW;

MatrixForm[A]
```

Out[191]//MatrixForm=

$$\begin{pmatrix} 0.6 & 0.17 & 0.52 & 0.09 & 0.02 \\ 3. & -0.51 & 0.13 & -0.27 & 0.04 \\ 1.8 & 0.85 & 0.26 & 0.45 & -0.01 \\ -3. & 0.34 & 0.26 & -0.18 & 0.02 \\ 1.2 & 0.34 & -0.39 & 0.36 & -0.04 \end{pmatrix}$$

(2) Checking consistency and uniqueness

To decide whether the system is consistent and whether the solution is unique, we check whether A is invertible .

In[192]:=

```
detU=Det[U]
detDW=Det[DW]
detA=Det[A]
```

Out[192]=

-78

Out[193]=

0.000011934

Out[194]=

-0.000930852

We have *A=U diag(w).*
Using the determinant property of matrix products, *det(A)= det(U) det(diag(w)).*
Since *det(U)=–78* which is not equal to 0, we know that the matrix U is invertible. Moreover, since w is a weight distribution, all entries satisfy w_i>0, this diagonal matrix is also invertible. Therefore A is invertible as the product of two invertible matrices.Hence the system $A\ \alpha = \tilde{w}$ is consistent for every target vector $\tilde{w}$ and the solution is unique.This does not depend on the particular choice or $\tilde{w}$, provided the entries of w stay nonzero.

(3) Solving for the realised tuning parameters

The realised tuning parameters satisfy $\boldsymbol{A\,\beta} = \tilde{\boldsymbol{w}}$. Since A is invertible, $\beta = \boldsymbol{A}^{-1}\ \tilde{\boldsymbol{w}}$

In the code, **wt** denotes the target weight distribution vector $\tilde{w}$ after transportation.

In[195]:=

```
beta = Inverse[A] . wt;
beta // MatrixForm
```

Out[196]//MatrixForm=

$$\begin{pmatrix} -1.52009 \\ -5.29035 \\ -12.3097 \\ 30.0499 \\ 294.897 \end{pmatrix}$$

so the realised tuning parameters are $\begin{pmatrix} -1.52009 \\ -5.29035 \\ -12.3097 \\ 30.0499 \\ 294.897 \end{pmatrix}$.

(4) Solving for the inputted tuning parameters

The engineers intended the transportation to preserve the population distribution . Hence they aimed for $\tilde{w}\ = w$

Therefore, the intended tuning parameters θ must satisfy the linear system $A\theta=w$

Since A is invertible, this system has a unique solution given by $\theta = \boldsymbol{A}^{-1}\ w$

In[197]:=

```
theta = Inverse[A].w;
theta // MatrixForm
```

Out[198]//MatrixForm=

$$\begin{pmatrix} -1.47051 \\ -5.33032 \\ -11.0118 \\ 28.7436 \\ 276.385 \end{pmatrix}$$

Hence the correct inputted tuning parameters are $\begin{pmatrix} -1.47051 \\ -5.33032 \\ -11.0118 \\ 28.7436 \\ 276.385 \end{pmatrix}$

(5) Modifying only u_1 to correct the transporter

To correct the transport error, we replace only the first column vector u_1 by a new vector v_1,while keeping the realised tuning parameters β unchanged. The modified system must satisfy :

$$\boldsymbol{w = \beta_1\, w_1\, v_1 + \beta_2\, w_2\, u_2 + \beta_3\, w_3\, u_3 + \beta_4\, w_4\, u_4 + \beta_5\, w_5\, u_5}$$

Then we have

$$v_1 = \frac{w - (\beta_2\, w_2\, u_2 + \beta_3\, w_3\, u_3 + \beta_4\, w_4\, u_4 + \beta_5\, w_5\, u_5)}{\beta_1\, w_1}$$

In[199]:=

```
v1 = (w - (beta[[2]]*w[[2]]*u2 +
        beta[[3]]*w[[3]]*u3 +
        beta[[4]]*w[[4]]*u4 +
        beta[[5]]*w[[5]]*u5)) / (beta[[1]]*w[[1]]);

v1 // MatrixForm
```

Out[200]//MatrixForm=

$$\begin{pmatrix} 0.76975 \\ 5.05482 \\ 3.01096 \\ -5.04386 \\ 2.20832 \end{pmatrix}$$

Verification

To verify the result, define the modified transport matrix A_{new}=[$w_1\, v_1$, $w_2\, u_2$, $w_3\, u_3$, $w_4\, u_4$, $w_5\, u_5$]
Then check that $A_{new}\beta$=w.

In[201]:=

```
Anew = Transpose[{w[[1]]*v1, w[[2]]*u2, w[[3]]*u3, w[[4]]*u4, w[[5]]*u5}];
Anew . beta // N
```

Out[202]=

{0.6, 0.17, 0.13, 0.09, 0.01}

Hence the corrected first column vector is $\begin{pmatrix} 0.76975 \\ 5.05482 \\ 3.01096 \\ -5.04386 \\ 2.20832 \end{pmatrix}$

MAR Part

(6) Analysing the Martian transporter

For Mars, the initial distribution is $\omega = (0.50, 0.20, 0.30)^T$, and the intrinsic vectors are $x_1 = (2, 4, 6)^T$, $x_2 = (1, 2, 3)^T$, $x_3 = (1, 2, 1)^T$.The Martian engineers want the transport to preserve the initial distribution, so they require $\tilde{\omega} = \omega$.The system can therefore be written in matrix form as ω = AMars α, where : AMars = X DMars.

In[203]:=

```
omega = {0.50, 0.20, 0.30};

x1 = {2, 4, 6};
x2 = {1, 2, 3};
x3 = {1, 2, 1};

X = Transpose[{x1, x2, x3}];
DMars = DiagonalMatrix[omega];
AMars = X . DMars;

MatrixForm[X]
MatrixForm[DMars]
MatrixForm[AMars]
```

Out[210]//MatrixForm=

$$\begin{pmatrix} 2 & 1 & 1 \\ 4 & 2 & 2 \\ 6 & 3 & 1 \end{pmatrix}$$

Out[211]//MatrixForm=

$$\begin{pmatrix} 0.5 & 0. & 0. \\ 0. & 0.2 & 0. \\ 0. & 0. & 0.3 \end{pmatrix}$$

Out[212]//MatrixForm=

$$\begin{pmatrix} 1. & 0.2 & 0.3 \\ 2. & 0.4 & 0.6 \\ 3. & 0.6 & 0.3 \end{pmatrix}$$

Checking whether the original Martian device works (Action i)

To determine whether action (i) works, we check whether the system $\boldsymbol{\omega} = \boldsymbol{A}_{\mathbf{Mars}}\, \boldsymbol{\alpha}$ is consistent.
Since $x_1 = (2, 4, 6)^T = 2(1, 2, 3)^T = 2\, x_2$,the columns of X are linearly dependent. Hence X and AMars are singular.
We confirm consistency using the augmented matrix.

In[213]:=

```
Det[X]
Det[AMars]
augMars = ArrayFlatten[{{AMars, Transpose[{omega}]}}];

RowReduce[augMars]
MatrixRank[AMars]
MatrixRank[augMars]
```

Out[213]=

```
0
```

Out[214]=

```
0.
```

Out[216]=

```
{{1, 0.2, 0., 0.}, {0, 0, 1, 0.}, {0, 0, 0, 1}}
```

Out[217]=

```
2
```

Out[218]=

```
3
```

Since det (X) = 0, the matrix X is singular, and hence AMars is singular . Row reduction of the augmented matrix shows that the system is inconsistent . Therefore there is no choice of tuning parameters $\alpha 1$, $\alpha 2$, $\alpha 3$ that makes the original Martian device preserve the initial distribution . Hence action (i) fails .

Redistributing the initial population (Action ii)

To determine whether action (ii) is possible, we must show that there exists a modified initial distribution ω (a probability vector with positive entries summing to 1) such that the transportation system $\mathrm{X}\,\mathrm{diag}(\omega')\alpha=\omega'$ has a solution.
This is equivalent to requiring that $\omega' \in \mathrm{col}(\mathrm{X})$, $\mathrm{col}\,(X) = \{\beta_1\, c_1+\beta_2\, c_2+\beta_3\, c_3\}$ since if $\omega'=\mathrm{X}\,\beta$ for some $\beta \in R^3$, then choosing $\alpha = \mathrm{diag}(\omega')^{-1}\beta$ satisfies the system.

Given X= $\begin{pmatrix} 2 & 1 & 1 \\ 4 & 2 & 2 \\ 6 & 3 & 1 \end{pmatrix}$, we observe that the second column equals half of the first, socol(X)=span{ c_1, c_3}

Thus any vector in the column space has form $\omega'=\begin{pmatrix} 2\,a+b \\ 4\,a+2\,b \\ 6\,a+b \end{pmatrix}$

Imposing the probability constraint $(\omega')_1+(\omega')_2+(\omega')_3=1$ gives 12a+4b=1

Choosing a=0, b=$\frac{1}{4}$ yields $\omega'=\begin{pmatrix} \frac{1}{4} \\ \frac{1}{2} \\ \frac{1}{4} \end{pmatrix}$ which has positive entries and lies in col(X).

We choose ω' proportional to c_3 for simplicity: $\omega'=\mathrm{X}\begin{pmatrix} 0 \\ 0 \\ \frac{1}{4} \end{pmatrix}$

Thus we may take $\beta=\begin{pmatrix} 0 \\ 0 \\ \frac{1}{4} \end{pmatrix}$, Now compute $\alpha = \mathrm{diag}(\omega')^{-1}\beta$,we get $\alpha = \begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix}$

Verify X $\operatorname{diag}(\omega^{'})\alpha$=X $\begin{pmatrix} 0 \\ 0 \\ \frac{1}{4} \end{pmatrix}$ $=\omega^{'}$

In[219]:=

```
X = {{2, 1, 1},
     {4, 2, 2},
     {6, 3, 1}};

wNew = {1/4, 1/2, 1/4};
alpha = {0, 0, 1};

X.DiagonalMatrix[wNew].alpha
```

Out[222]=

$\left\{\frac{1}{4}, \frac{1}{2}, \frac{1}{4}\right\}$

Therefore action (ii) succeeds . One valid modified distribution is $\boldsymbol{\omega'} = (\frac{1}{4}, \frac{1}{2}, \frac{1}{4})^T$, with tuning parameters $\alpha = (0, 0, 1)^T$

Repairing or rebuilding the device (Action iii)

To repair the transporter, we replace the original device vectors x1, x2 and x3 by new vectors forming a matrix $X^{'}$,The transportation condition becomes $X^{'}$ diag(ω) $\alpha = \omega$
Let D=diag(ω) Then $X^{'}$ (D α) = ω
If $X^{'}$ is invertible, this system always has a unique solution $\alpha = D^{-1} \left(X^{'}\right)^{-1} \omega$
since all entries of ω are positive and hence D^{-1}exists and α is uniquely determined.
Therefore, rebuilding the device succeeds by choosing new vectors so that $X^{'}$ has full rank.
A particularly simple and natural repair is to choose $X^{'} = I_3$
I $\operatorname{diag}(\omega)\alpha=\omega \quad \Longrightarrow \quad \operatorname{diag}(\omega)\alpha=\omega$.
Multiplying by D^{-1} on the left gives $\alpha = \begin{pmatrix} 1 \\ 1 \\ 1 \end{pmatrix}$
This corresponds to rebuilding the transporter using the standard basis vectors
x1' = $(1, 0, 0)^T$, x2' = $(0, 1, 0)^T$, x3' = $(0, 0, 1)^T$.
Hence the rebuilt device preserves the original Martian population distribution exactly, and action (iii) succeeds.

In[88]:=

```
w = {w1, w2, w3};
Xnew = IdentityMatrix[3];
alpha = {1, 1, 1};

Xnew.DiagonalMatrix[w].alpha
```

Out[91]=

```
{w1, w2, w3}
```

Thus the rebuilt device preserves the original Martian distribution exactly . Therefore action (iii) succeeds .

(7) Testing the new Martian device

The redesigned Martian transporter satisfies $\tilde{\omega} = B\alpha$ **where B** $= [w_1\, y_1, w_2\, y_2, w_3\, y_3, (w_1\, w_2\, w_3)\, y_4]$

The given vectors are $y_1 = \begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix}$, $y_2 = \begin{pmatrix} 4 \\ 8 \\ -3 \end{pmatrix}$, $y_3 = \begin{pmatrix} 1 \\ 2 \\ 2 \end{pmatrix}$,and the two candidate vectors are $y_{4A} = \begin{pmatrix} 1 \\ 2 \\ 4 \end{pmatrix}$, $y_{4B} = \begin{pmatrix} 1 \\ 4 \\ 2 \end{pmatrix}$, We test both possibilities to determine which choice gives a consistent system with target distribution $\tilde{\omega} = \omega$.

In[92]:=
```
y1 = {1, 2, 3};
y2 = {4, 8, -3};
y3 = {1, 2, 2};
y4A = {1, 2, 4};
y4B = {1, 4, 2};
omegaProd = Times @@ omega
```

Out[97]=
```
0.03
```

Case A : $y_{4A} = (1, 2, 4)^T$

The coefficient matrix is formed by the columns $\omega_1\, y_1$, $\omega_2\, y_2$, $\omega_3\, y_3$, $\omega_1\, \omega_2\, \omega_3\, y_{4A}$

In[98]:=
```
BA = Transpose[{omega[[1]] y1, omega[[2]] y2, omega[[3]] y3, omegaProd y4A}];

MatrixForm[BA]

augA = ArrayFlatten[{{BA, Transpose[{omega}]}}];
rowA = RowReduce[augA]
```

Out[99]//MatrixForm=

$$\begin{pmatrix} 0.5 & 0.8 & 0.3 & 0.03 \\ 1. & 1.6 & 0.6 & 0.06 \\ 1.5 & -0.6 & 0.6 & 0.12 \end{pmatrix}$$

Out[101]=
```
{{1, 0., 0.44, 0.076, 0.}, {0, 1, 0.1, -0.01, 0.}, {0, 0, 0, 0, 1}}
```

Row reduction of the augmented matrix shows that the system is inconsistent. And a quick structural observation is that all four columns satisfy second entry = 2*first entry .But the target vector $\omega = (0.5,\ 0.2,\ 0.3)^T$ does not, because 0.2≠2*(0.5) So no solution exists.

Case B:$y_{4B} = (1, 4, 2)^T$

The coefficient matrix is formed by the columns $\omega_1\, y_1$, $\omega_2\, y_2$, $\omega_3\, y_3$, $\omega_1\, \omega_2\, \omega_3\, y_{4B}$

In[102]:=

```
BB = Transpose[{omega[[1]] y1, omega[[2]] y2, omega[[3]] y3, omegaProd y4B}];

MatrixForm[BB]

augB = ArrayFlatten[{{BB, Transpose[{omega}]}}];
rowB = RowReduce[augB]
```

Out[103]//MatrixForm=

$$\begin{pmatrix} 0.5 & 0.8 & 0.3 & 0.03 \\ 1. & 1.6 & 0.6 & 0.12 \\ 1.5 & -0.6 & 0.6 & 0.06 \end{pmatrix}$$

Out[105]=

```
{{1, 0., 0.44, 0., 0.946667}, {0, 1, 0.1, 0., 0.533333}, {0, 0, 0, 1, -13.3333}}
```

Row reduction shows that this system is consistent, so this choice of y_4 works . To obtain one explicit solution, we set the free parameter equal to 0.

In[106]:=

```
alphaB = {71/75, 8/15, 0, -40/3};

checkB = BB . alphaB
Chop[checkB - omega]
N[alphaB, 6]
```

Out[107]=

```
{0.5, 0.2, 0.3}
```

Out[108]=

```
{0, 0, 0}
```

Out[109]=

```
{0.946667, 0.533333, 0, -13.3333}
```

The result equals ω, so the choice $y_4 = (1, 4, 2)^T$ works .

3. Summary

Earth part

(a)

$A=[w_1\ u_1\ w_2\ u_2\ w_3\ u_3\ w_4\ u_4\ w_5\ u_5]$, so $A=U\ DiagonalMatrix[w]$

$$A=\begin{pmatrix} 0.6 & 0.17 & 0.52 & 0.09 & 0.02 \\ 3. & -0.51 & 0.13 & -0.27 & 0.04 \\ 1.8 & 0.85 & 0.26 & 0.45 & -0.01 \\ -3. & 0.34 & 0.26 & -0.18 & 0.02 \\ 1.2 & 0.34 & -0.39 & 0.36 & -0.04 \end{pmatrix}$$

(b)

The system A $\alpha=\tilde{w}$ is consistent and has a unique solution, because A is invertible. This holds for any target vector $\tilde{w}$, and for any valid weight distribution w (with all entries positive). This does not depend on the particular choice or $\tilde{w}$, provided the entries of w stay nonzero.

(c)

The realised tuning parameters are $\begin{pmatrix} -1.52009 \\ -5.29035 \\ -12.3097 \\ 30.0499 \\ 294.897 \end{pmatrix}$

(d)

The correct inputted tuning parameters are $\begin{pmatrix} -1.47051 \\ -5.33032 \\ -11.0118 \\ 28.7436 \\ 276.385 \end{pmatrix}$

(e)

$v_1 = \begin{pmatrix} 0.76975 \\ 5.05482 \\ 3.01096 \\ -5.04386 \\ 2.20832 \end{pmatrix}$

This is verified by checking that the modified system satisfies $A_{new}\beta$=w

Mars part

(f) :

Action (i) fails

Action (ii) succeeds

Action (iii) succeeds

(f) (i)

The original Martian device does not work . The system is inconsistent, so no tuning parameters a_1, a_2, a_3 can make $\tilde{\omega}$=ω

(f)(ii)

Redistributing the population can work. One valid choice is $w' = (\frac{1}{4}, \frac{1}{2}, \frac{1}{4})^T$, $\alpha = (0, 0, 1)^T$ for which X diag(w') α=w'

(f)(iii)

Repairing the device also can work. For example, choosing x1’ = $(1, 0, 0)^T$, x2’ = $(0, 1, 0)^T$, x3’ = $(0, 0, 1)^T$ with $\alpha = (1, 1, 1)^T$

(g)

For the new device, test both candidates:

If y_4=$\begin{pmatrix} 1 \\ 2 \\ 4 \end{pmatrix}$, the system is inconsistent, so this choice does not work.

If y_4=$\begin{pmatrix} 1 \\ 4 \\ 2 \end{pmatrix}$, the system is consistent, so this choice does work.

Hence the correct choice is y_4=$\begin{pmatrix} 1 \\ 4 \\ 2 \end{pmatrix}$.

Problem 3 : Determinants and n - dimensional volume

1. Problem Restatement

We investigate the relationship between determinants and n - dimensional volume . For n vectors $z_1,...,z_n \in R^m$, let Z be the m×n matrix with these vectors as columns.The Gram matrix is $G_Z = Z^T Z$. The n-vol of the parallelepiped they span is defined as n-vol= $\sqrt{\det(G_Z)}$. When m=n, this reduces to $|\det(G_Z)|$.

We are given two sets of vectors :

Set A (in R^5) :

$$a_1 = \begin{pmatrix} 2 \\ -1 \\ 3 \\ 1 \\ 4 \end{pmatrix}, \quad a_2 = \begin{pmatrix} -2 \\ 5 \\ 3 \\ 5 \\ 4 \end{pmatrix}, \quad a_3 = \begin{pmatrix} -1 \\ 4 \\ 2 \\ 1 \\ 2 \end{pmatrix}$$

Set B (in R^3) :

$$b_1 = \begin{pmatrix} 1 \\ 3 \\ 5 \end{pmatrix}, \quad b_2 = \begin{pmatrix} 2 \\ 6 \\ 10 \end{pmatrix}, \quad b_3 = \begin{pmatrix} -3 \\ 2 \\ 3 \end{pmatrix}, \quad b_4 = \begin{pmatrix} 4 \\ 3 \\ 1 \end{pmatrix}, \quad b_5 = \begin{pmatrix} 1 \\ -2 \\ -4 \end{pmatrix}$$

The objectives are:

(a) compute the 3-vol of the parallelepiped spanned by a_1, a_2, a_3

(b) compute the 2-vol of the parallelogram spanned by a_1, a_3

(c) obtain the same 2-vol using a suitable submatrix of the Gram matrix

(d) compute the determinant of the Gram matrix G_B and interpret the result geometrically

(e) compute the 3-vol of the parallelepiped spanned by b_2, b_4, b_5

(f) determine the largest integer n such that a subset of $\{b_1, b_2, b_3\}$ spans a geometric object with non-zero n-vol.

2. Methodology and Calculations

(1) Define the Vectors and Gram Matrices

In[110]:=

```
ClearAll["Global`*"]

a1 = {2, -1, 3, 1, 4};
a2 = {-2, 5, 3, 5, 4};
a3 = {-1, 4, 2, 1, 2};

A = Transpose[{a1, a2, a3}];
GA = Transpose[A] . A;

b1 = {1, 3, 5};
b2 = {2, 6, 10};
b3 = {-3, 2, 3};
b4 = {4, 3, 1};
b5 = {1, 4, -2};

B = Transpose[{b1, b2, b3, b4, b5}];
GB = Transpose[B] . B;
A // MatrixForm
GA // MatrixForm

B // MatrixForm
GB // MatrixForm
```

Out[123]//MatrixForm=

$$\begin{pmatrix} 2 & -2 & -1 \\ -1 & 5 & 4 \\ 3 & 3 & 2 \\ 1 & 5 & 1 \\ 4 & 4 & 2 \end{pmatrix}$$

Out[124]//MatrixForm=

$$\begin{pmatrix} 31 & 21 & 9 \\ 21 & 79 & 41 \\ 9 & 41 & 26 \end{pmatrix}$$

Out[125]//MatrixForm=

$$\begin{pmatrix} 1 & 2 & -3 & 4 & 1 \\ 3 & 6 & 2 & 3 & 4 \\ 5 & 10 & 3 & 1 & -2 \end{pmatrix}$$

Out[126]//MatrixForm=

$$\begin{pmatrix} 35 & 70 & 18 & 18 & 3 \\ 70 & 140 & 36 & 36 & 6 \\ 18 & 36 & 22 & -3 & -1 \\ 18 & 36 & -3 & 26 & 14 \\ 3 & 6 & -1 & 14 & 21 \end{pmatrix}$$

The matrix A contains the vectors a_1, a_2, a_3 as columns, and $G_A = A^T A$ is its Gram matrix . Similarly, the matrix B contains the vectors b_1, b_2, b_3, b_4, b_5 as columns, and $G_B = B^T B$ is its Gram matrix .

(2) 3 - vol spanned by a_1, a_2, a_3

The vectors a_1,a_2,a_3 lie in R^5, so matrix A is a 5×3 matrix.The relevant Gram matrix isThe Gram matrix G_A= A^T A. The 3-dimensional volume is therefore 3-vol= $\sqrt{\det(G_A)}$

In[127]:=

```
Det[GA]
volA=FullSimplify[Sqrt[Det[GA]]]
```

Out[127]=

9196

Out[128]=

$22\sqrt{19}$

Therefore, the 3 - vol is $22\sqrt{19}$

(3) 2 - vol spanned by a_1, a_3

To find the 2 - dimensional volume spanned by a_1 and a_3, we form the 5×2 matrix with columns a_1 and a_3, Its Gram matrix is $\begin{pmatrix} a_1\,a_1 & a_1\,a_3 \\ a_3\,a_1 & a_3\,a_3 \end{pmatrix}$.

Then the 2-vol is 2-vol= $\sqrt{\det(G)}$

In[129]:=

```
G13={{a1.a1,a1.a3},{a3.a1,a3.a3}};
G13//MatrixForm
Det[G13]
vol13=FullSimplify[Sqrt[Det[G13]]]
```

Out[130]//MatrixForm=

$$\begin{pmatrix} 31 & 9 \\ 9 & 26 \end{pmatrix}$$

Out[131]=

725

Out[132]=

$5\sqrt{29}$

Therefore, the area of the parallelogram spanned by a_1 and a_3 is $5\sqrt{29}$

(4) The same 2 - vol using a submatrix of G_A

Since G_A is the Gram matrix of the ordered set (a_1, a_2, a_3), the vectors a_1 and a_3 correspond to rows and columns 1 and 3. Therefore, to obtain the Gram matrix for a_1 and a_3 from G_A, we delete row 2 and column 2.

In[133]:=

```
subGA=GA[[{1,3},{1,3}]];
subGA//MatrixForm
Det[subGA]
volSubGA=FullSimplify[Sqrt[Det[subGA]]]
```

Out[134]//MatrixForm=

$$\begin{pmatrix} 31 & 9 \\ 9 & 26 \end{pmatrix}$$

Out[135]=

725

Out[136]=

$5\sqrt{29}$

The required submatrix is $\begin{pmatrix} 31 & 9 \\ 9 & 26 \end{pmatrix}$. Its determinant is again 725, and hence 2 - vol = $\sqrt{725}$ = 5 $\sqrt{29}$. This agrees with the answer in part (b), as expected .

(5) Compute det (G_A) and interpret geometrically

Now we consider the five vectors b_1, b_2, b_3, b_4, b_5 in $\mathbb{R}^3$. Their Gram matrix is $G_B = B^T B$. We compute its determinant directly .

In[137]:=

```
GB//MatrixForm
Det[GB]
MatrixRank[B]
```

Out[137]//MatrixForm=

$$\begin{pmatrix} 35 & 70 & 18 & 18 & 3 \\ 70 & 140 & 36 & 36 & 6 \\ 18 & 36 & 22 & -3 & -1 \\ 18 & 36 & -3 & 26 & 14 \\ 3 & 6 & -1 & 14 & 21 \end{pmatrix}$$

Out[138]=

0

Out[139]=

3

Since B is a 3×5 matrix, we have rank(B)≤3. Hence rank(B^TB)≤3<5.Therefore G_B is singular, so det(G_B)=0. Also, $b_2 = 2\, b_1$, so the vectors are linearly dependent, which is another reason why the Gram determinant is zero.

The determinant of G_B is 0.This agrees with the geometric interpretation . The matrix B has five column vectors in $\mathbb{R}^3$, so at most three of them can be linearly independent . Therefore, five vectors in $\mathbb{R}^3$ cannot span a genuine 5 - dimensional parallelepiped with non - zero 5 - vol . Hence the corresponding generalised volume must be zero, and this is exactly reflected by det (G_B) = 0.

(6) 3 - vol spanned by b_2, b_4, b_5

To compute the 3 - vol of the parallelepiped spanned by b_2, b_4, b_5, we use the corresponding submatrix of G_B . Because these vectors are the 2 nd, 4 th and 5 th columns of B, we keep rows and columns 2, 4 and 5 of G_B .

In[140]:=

```
subGB245=GB[[{2,4,5},{2,4,5}]];
subGB245//MatrixForm
Det[subGB245]
vol245=FullSimplify[Sqrt[Det[subGB245]]]
```

Out[141]//MatrixForm=

$$\begin{pmatrix} 140 & 36 & 6 \\ 36 & 26 & 14 \\ 6 & 14 & 21 \end{pmatrix}$$

Out[142]=

26 896

Out[143]=

164

The required submatrix is $\begin{pmatrix} 140 & 36 & 6 \\ 36 & 26 & 14 \\ 6 & 14 & 21 \end{pmatrix}$. Its determinant is 26896, so 3 - vol = $\sqrt{26\,896}$ =$116\sqrt{2}$,$116\sqrt{2}$≈164. Therefore, the 3 - dimensional volume of the parallelepiped spanned by b_2, b_4, b_5 is 164.

(7) Largest n for which a subset of {b_1, b_2, b_3} has non - zero n - vol

We now determine the largest integer n such that some subset of {b_1, b_2, b_3} spans a geometric object with non - zero n - vol . First observe that $b_2 = 2\, b_1$. So b_1 and b_2 are linearly dependent . Therefore, the set {b_1, b_2, b_3} cannot contain three linearly independent vectors, and hence no subset of these three vectors can have non - zero 3 - vol . However, the pair {b_1, b_3} is linearly independent, so it should have non - zero 2 - vol . We verify this by taking the corresponding 2*2 submatrix of G_B .

In[144]:=

```
subGB13=GB[[{1,3},{1,3}]];
subGB13//MatrixForm
Det[subGB13]
volb1b3=FullSimplify[Sqrt[Det[subGB13]]]
subGB123=GB[[{1,2,3},{1,2,3}]];
subGB123//MatrixForm
Det[subGB123]
```

Out[145]//MatrixForm=

$$\begin{pmatrix} 35 & 18 \\ 18 & 22 \end{pmatrix}$$

Out[146]=

446

Out[147]=

$\sqrt{446}$

Out[149]//MatrixForm=

$$\begin{pmatrix} 35 & 70 & 18 \\ 70 & 140 & 36 \\ 18 & 36 & 22 \end{pmatrix}$$

Out[150]=

0

For the pair {b_1, b_3}, the relevant submatrix has determinant 446, so the 2 - vol is $\sqrt{446}$,which is not equal to 0.But for the triple {b_1, b_2, b_3}, the determinant of the corresponding 3*3 submatrix is 0, so the 3 - vol is zero . Therefore, the largest integer n for which there exists a subset of $\{b_1, b_2, b_3\}$ with non - zero n - vol is n = 2.

3. Summary

(a)

The 3-dimensional volume of the parallelepiped spanned by a_1,a_2,a_3 is 3-vol= $22\sqrt{19} \approx 95.8968$

(b)

The 2-dimensional volume (area) of the parallelogram spanned by a1 and a3 is 2-vol = $5\sqrt{29} \approx 26.9258$

(c)

Using a principal submatrix of the Gram matrix G_A, the same 2-dimensional volume was obtained, confirming consistency with part (b)

(d)

The determinant of the Gram matrix G_B is det $(G_B) = 0$

This aligns with geometric intuition, since five vectors in R^3 cannot span a 5-dimensional geometric object

(e)

The 3-dimensional volume of the parallelepiped spanned by b_2, b_4, b_5 is 3-vol = 164.

(f)

Since b_2 is linearly dependent on b_1, the set $\{b_1, b_2, b_3\}$ contains at most two linearly independent vectors. Therefore, the largest integer n for which there exists a subset with non-zero n-dimensional volume is n = 2